<commit_before># Adds version control information to the variable VERS. For
# determining the Version Control System used (if any) it inspects the
# existence of certain subdirectories under CMAKE_CURRENT_SOURCE_DIR.

function(add_version_info_from_vcs VERS)
  set(result ${${VERS}})
  if( EXISTS "${CMAKE_CURRENT_SOURCE_DIR}/.svn" )
    set(result "${result}svn")
    # FindSubversion does not work with symlinks. See PR 8437
    if( NOT IS_SYMLINK "${CMAKE_CURRENT_SOURCE_DIR}" )
      find_package(Subversion)
    endif()
    if( Subversion_FOUND )
      subversion_wc_info( ${CMAKE_CURRENT_SOURCE_DIR} Project )
      if( Project_WC_REVISION )
        set(result "${result}-r${Project_WC_REVISION}")
      endif()
    endif()
  elseif( EXISTS ${CMAKE_CURRENT_SOURCE_DIR}/.git )
    set(result "${result}git")
    # Try to get a ref-id
    find_program(git_executable NAMES git git.exe git.cmd)
    if( git_executable )
      execute_process(COMMAND ${git_executable} show-ref HEAD
                      WORKING_DIRECTORY ${CMAKE_CURRENT_SOURCE_DIR}
                      TIMEOUT 5
                      RESULT_VARIABLE git_result
                      OUTPUT_VARIABLE git_output)
      if( git_result EQUAL 0 )
        string(SUBSTRING ${git_output} 0 7 git_ref_id)
        set(result "${result}-${git_ref_id}")
      else()
        execute_process(COMMAND ${git_executable} svn log --limit=1 --oneline
                        WORKING_DIRECTORY ${CMAKE_CURRENT_SOURCE_DIR}
                        TIMEOUT 5
                        RESULT_VARIABLE git_result
                        OUTPUT_VARIABLE git_output)
        if( git_result EQUAL 0 )
          string(REGEX MATCH r[0-9]+ git_svn_rev ${git_output})
          set(result "${result}-svn-${git_svn_rev}")
        endif()
      endif()
    endif()
  endif()
  set(${VERS} ${result} PARENT_SCOPE)
endfunction(add_version_info_from_vcs)<commit_msg>At the request of <NAME>, make the VCS version detection logic in CMake a bit more handy. Previously we would get such charming versions as the following for revision NNNN and commit-ish XXXXX:   3.1svnsvn-rNNNN   3.1svngit-svn-rNNNN   3.1svngit-svn-XXXXX<commit_after># Adds version control information to the variable VERS. For
# determining the Version Control System used (if any) it inspects the
# existence of certain subdirectories under CMAKE_CURRENT_SOURCE_DIR.

function(add_version_info_from_vcs VERS)
  string(REPLACE "svn" "" result "${${VERS}}")
  if( EXISTS "${CMAKE_CURRENT_SOURCE_DIR}/.svn" )
    set(result "${result}svn")
    # FindSubversion does not work with symlinks. See PR 8437
    if( NOT IS_SYMLINK "${CMAKE_CURRENT_SOURCE_DIR}" )
      find_package(Subversion)
    endif()
    if( Subversion_FOUND )
      subversion_wc_info( ${CMAKE_CURRENT_SOURCE_DIR} Project )
      if( Project_WC_REVISION )
        set(result "${result}-r${Project_WC_REVISION}")
      endif()
    endif()
  elseif( EXISTS ${CMAKE_CURRENT_SOURCE_DIR}/.git )
    set(result "${result}git")
    # Try to get a ref-id
    find_program(git_executable NAMES git git.exe git.cmd)
    if( git_executable )
      set(is_git_svn_rev_exact false)
      execute_process(COMMAND ${git_executable} svn log --limit=1 --oneline
                      WORKING_DIRECTORY ${CMAKE_CURRENT_SOURCE_DIR}
                      TIMEOUT 5
                      RESULT_VARIABLE git_result
                      OUTPUT_VARIABLE git_output)
      if( git_result EQUAL 0 )
        string(REGEX MATCH r[0-9]+ git_svn_rev ${git_output})
        string(SUBSTRING "${git_svn_rev}" 1 -1 git_svn_rev_number)
        set(git_svn_rev "-svn-${git_svn_rev}")

        # Determine if the HEAD points directly at a subversion revision.
        execute_process(COMMAND ${git_executable} svn find-rev HEAD
                        WORKING_DIRECTORY ${CMAKE_CURRENT_SOURCE_DIR}
                        TIMEOUT 5
                        RESULT_VARIABLE git_result
                        OUTPUT_VARIABLE git_output)
        if( git_result EQUAL 0 )
          string(STRIP "${git_output}" git_head_svn_rev_number)
          if( git_head_svn_rev_number EQUAL git_svn_rev_number )
            set(is_git_svn_rev_exact true)
          endif()
        endif()
      else()
        set(git_svn_rev "")
      endif()
      execute_process(COMMAND
                      ${git_executable} show-ref --abbrev --hash --head HEAD
                      WORKING_DIRECTORY ${CMAKE_CURRENT_SOURCE_DIR}
                      TIMEOUT 5
                      RESULT_VARIABLE git_result
                      OUTPUT_VARIABLE git_output)
      if( git_result EQUAL 0 AND NOT is_git_svn_rev_exact )
        string(STRIP "${git_output}" git_ref_id)
        set(result "${result}${git_svn_rev}-${git_ref_id}")
      else()
        set(result "${result}${git_svn_rev}")
      endif()
    endif()
  endif()
  set(${VERS} ${result} PARENT_SCOPE)
endfunction(add_version_info_from_vcs)
{'fraction_non_alphanumeric': 0.08915123142682678, 'fraction_numerical': 0.006920415224913495, 'mean_word_length': 2.5660377358490565, 'pattern_counts': {'":': 0, '<': 4, '<?xml version=': 0, '>': 4, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '2212123', 'n_tokens_mistral': 1629, 'n_tokens_neox': 1523, 'n_words': 361}